
Controlling DC Motors 

Objective   
The aim of this lab experiment is to  control a small DC motor. 

Equipment 
Flight 8086 training board, Application board, PC with Flight86 software, download 
cable 

Tasks to be Performed 
 Running the motor in forward and reverse direction for a specified time 

 Controlling the speed of the motor 

 

 

 

 

 

 

 

 

 

 

Experiment 

5.1 DC Motor 
The Application Board contains a small DC motor that can be operated in the forward 
or reverse direction. The operation of this DC motor is controlled by bits 6 and 7 on 
Port-A as shown in Table 5.1. 

 
Table 5.1: Operation Modes of the DC Motor 

Bit6    Bit7 Operation 
0         0 
0         1 
1         0 
1         1 

Stop 
Reverse Direction 
Forward Direction 

Stop 
 
The following example shows you how to run the DC motor in the forward and 
reverse direction for a specific time. 

 
 

Example 4.1: Write a program to run the DC motor in the forward direction for 
5 seconds, turn it off for 3 seconds, then run it in the revere direction for 5 
seconds.  
 
 
Set SW2-1 to SWITCH 
Set SW2-2 to MOTOR 
SW4-1, SW4-2, SW4-3, and SW4-4  OFF 
 
MOV   AL, 99h    ; initialize the 8255 PPI chip 
OUT   06h, AL    ; A input, B output, C input 
MOV   DL,  ?   ; load a proper value for 5s delay 
MOV   AL, 40h  ; forward direction 
OUT   02h, AL 
CALL Delay 
MOV   DL, ?    ; load a proper value for 3s delay 
MOV   AL, 00h    ; stop the motor 
OUT   02h, AL 
CALL  Delay 
MOV DL, ?    ; load a proper value for 5s delay 
MOV AL, 80h    ; reverse direction 
OUT 02h, AL 
CALL Delay 
MOV AL, 00h    ; stop the motor 
OUT 02h, AL 
INT 5 
 
; the delay procedure is left as an exercise 

5.3 Controlling the Speed of the DC Motor 
When the DC motor is ON (forward/reverse), it operates in its maximum speed. 
However, the speed of the motor can be controlled using pulse width modulation (PWM).  

PWM is a common technique for speed control. A good analogy is bicycle riding. You 
peddle (exert energy) and then coast (relax) using your momentum to carry you 
forward. As you slow down (due to wind resistance, friction, road shape) you peddle to 
speed up and then coast again. The duty cycle is the ratio of peddling time to the total 
time (peddle+coast time). A 100% duty cycle means you are peddling all the time, and 
50% only half the time.  

PWM for motor speed control works in a very similar way. Instead of peddling, your 
motor is given a fixed voltage value (turned on) and starts spinning. The voltage is then 
removed (turned off) and the motor "coasts". By continuing this voltage on-off duty 
cycle, motor speed is controlled.  

The concept of PWM inherently requires timing. The 8253 PIT chip can be used to 
generate PWM. In the beginning, the motor is turned on and Counter 0 is loaded with 
the ON duration. When Counter 0 terminates, the motor is turned off and Counter 1 
is loaded with the OFF duration. Now, when Counter 1 terminates, the process is 
repeated from the beginning. 

 

Example 4.2: Write a program to control the speed of the DC motor based on 
the state of Bit0 of the DIP switch. If Bit0 = 0, the motor will run at maximum 
speed. Otherwise, it will run at 50% of its duty cycle. 

12 
13 
14 
15 

 
Set SW2-1 to SWITCH 
Set SW2-2 to MOTOR 
SW4-1, SW4-2, SW4-3, and SW4-4  OFF 
COMSEG SEGMENT BYTE PUBLIC 'CODE' 
ASSUME CS:COMSEG, DS:COMSEG, ES:COMSEG, SS:COMSEG 
ORG 0100h 
Start:         ; set the external segment to point to the 
       ; base of the Interrupt Vector Table (IVR) 
  XOR AX,AX 
  MOV ES,AX 
 
       ;store the offset of ISR in the IVT 
       MOV WORD PTR ES:[38*4],OFFSET IR6_ROUTINE 
       ;store the segment of ISR in the IVT  
  MOV WORD PTR ES:[38*4+2],CS                 
 
       ;store the offset of ISR in the IVT 
       MOV WORD PTR ES:[39*4],OFFSET IR7_ROUTINE 
       ;store the segment of ISR in the IVT  
  MOV WORD PTR ES:[39*4+2],CS                 

       ; initialize the 8255 PPI chip: 
       ; A and C input ports, B output port 
  MOV AL, 99h 
  OUT 06h, AL 
    
       ; initialize the 8259 PIC chip 
  MOV AL, 17h        
       OUT 10h, AL 
  MOV AL, 20h 
  OUT 12h, AL 
  MOV AL, 03h 
  OUT 12h, AL 
  MOV AL, 3Fh 
  OUT 12h, AL 
 
       ; initialize 8253 PIT chip  (00110000 = 30h) 
       ; Counter0, load MSB then LSB, mode 0, binary  
  MOV AL, 30h 
  OUT 0Eh, AL 
  ; initialize 8253 PIT chip  (01110000 = 70h) 
       ; Counter1, load MSB then LSB, mode 0, binary 
       MOV AL, 70h 
  OUT 0Eh, AL 
 
       ; counter0 loaded with FFFFh 
  MOV AL, 0FFh     
       OUT 08h, AL  ; first load low byte 
  MOV AL, 0FFh     
  OUT 08h, AL  ; now load high byte 
 
  STI          ; enable 8086 maskable interrupts  
           
       ; start of main program 
     MOV AL, 40h ; turn on the motor 
  OUT 02h, AL 
again: JMP again   ; wait for interrupt on IR6/IR7  
                   ; Counter0/Counter1 decrements to 0  
 
; Interrupt Service Routine (ISR) for IR6 
; this routine checks Bit0 of the DIP switch 
; If Bit0 = 0 continue running the motor (max speed) 
; If Bit0 = 1 stop the motor (50% duty cycle)  
; the routine also reload Counter 1 
 
IR6_ROUTINE: 
    IN   AL, 00h  ; read DIP switch 
  
    TEST AL, 01h  ; check Bit0 
    JZ   continue ; if Bit0=0 then don't stop the motor 
    MOV  AL, 00h  ; else stop the motor   
    OUT  02h, AL  
 
continue: 
    ; counter1 loaded with FFFFh (50% duty cycle) 
    MOV AL, 0FFh     

    OUT 0Ah, AL   ; first load low byte 
    MOV AL, 0FFh     
    OUT 0Ah, AL   ; now load high byte 
    IRET 
 
; Interrupt Service Routine (ISR) for IR7 
; this routine turn on the motor and reload Counter 0 
 
IR7_ROUTINE: 
            
    MOV AL,  40h  ; turn on the motor 
    OUT 02h, AL 
            
    ; counter0 loaded with FFFFh 
    MOV AL, 0FFh     
         OUT 08h, AL   ; first load low byte 
    MOV AL, 0FFh     
    OUT 08h, AL   ; now load high byte 
 
    IRET 
 
COMSEG  ENDS 
END     start 
 

 

Exercises 
5.1. Modify Example 5.2 to allow the user to control the direction in addition to 

the speed (Use Bit1 to control the direction). 

5.2. Modify Example 5.2 to operate the motor at 4 different, for example 100% 
duty cycle, 50% duty cycle, 25% duty cycle, and 5 % duty cycle. The speed 
is selected based on the states of Bit0 and Bit1. 

 

 

 

 

 

 

 

 


